

1 ANSYS 概述

1.1 ANSYS 简介

ANSYS 是一种广泛的商业套装工程分析软件。所谓工程分析软件，主要是在机械结构系统受到外力负载所出现的反应，例如应力、位移、温度等，根据该反应可知道机械结构系统受到外力负载后的状态，进而判断是否符合设计要求。一般机械结构系统的几何结构相当复杂，受的负载也相当多，理论分析往往无法进行。想要解答，必须先简化结构，采用数值模拟方法分析。由于计算机行业的发展，相应的软件也应运而生，ANSYS 软件在工程上应用相当广泛，在机械、电机、土木、电子及航空等领域的使用，都能达到某种程度的可信度，颇获各界好评。使用该软件，能够降低设计成本，缩短设计时间。

到 80 年代初期，国际上较大型的面向工程的有限元通用软件主要有：ANSYS, NASTRAN, ASKA, ADINA, SAP 等。以 ANSYS 为代表的工程数值模拟软件，是一个多用途的有限元法分析软件，它从 1971 年的 2.0 版本与今天的 5.7 版本已有很大的不同，起初它仅提供结构线性分析和热分析，现在可用来求结构、流体、电力、电磁场及碰撞等问题的解答。它包含了前置处理、解题程序以及后置处理，将有限元分析、计算机图形学和优化技术相结合，已成为现代工程学问题必不可少的有力工具。

1.2 ANSYS 软件主要功能

ANSYS 软件是融结构、热、流体、电磁、声学于一体的大型通用有限元软件，可广泛的用于核工业、铁道、石油化工、航空航天、机械制造、能源、汽车交通、国防军工、电子、土木工程、生物医学、水利、日用家电等一般工业及科学研究。该软件提供了不断改进的功能清单，具体包括：结构高度非线性分析、电磁分析、计算流体力学分析、设计优化、接触分析、自适应网格划分及利用 ANSYS 参数设计语言扩展宏命令功能。

1.3 ANSYS 软件主要特点

主要技术特点：

- 唯一能实现多场及多场耦合分析的软件
- 唯一实现前后处理、求解及多场分析统一数据库的一体化大型 FEA 软件
- 唯一具有多物理场优化功能的 FEA 软件
- 唯一具有中文界面的大型通用有限元软件
- 强大的非线性分析功能
- 多种求解器分别适用于不同的问题及不同的硬件配置
- 支持异种、异构平台的网络浮动，在异种、异构平台上用户界面统一、数据文件全部兼容
- 强大的并行计算功能支持分布式并行及共享内存式并行
- 多种自动网格划分技术
- 良好的用户开发环境

支持的图形传递标准：

- SAT
- Parasolid

- STEP

与 CAD 软件的接口

- Unigraphics
- Pro/ENGINEER
- I-Deas
- Catia
- CADD5
- SolidEdge
- SolidWorks

1.4 运行环境 (ANSYS5.7)

Computer: Pentium-class system

Memory (RAM): 64 MB以上

Hard Disk: 500MB以上自由空间.

Operating System: Microsoft Windows 2000, Windows NT 4.0 (SP 5 or higher) or Windows 98

Graphics: A Windows 2000/NT 4.0 or 98 supported Graphics Card, capable of 1024x768 in High Color (16-bit). A 17 inch monitor (or larger) compatible with the above mentioned card is recommended.

2 ANSYS 的基本使用

2.1 ANSYS环境简介

ANSYS有两种模式：一种是交互模式 (Interactive Mode)，另一个是非交互模式 (Batch Mode)。交互模式是初学者和大多数使用者所采用，包括建模、保存文件、打印图形及结果分析等，一般无特别原因皆用交互模式。但若分析的问题要很长时间，如一、两天等，可把分析问题的命令做成文件，利用它的非交互模式进行分析。

运行该程序一般采用 Interactive 进入，这样可以定义工作名称，并且存放到指定的工作目录中。若使用 Run Interactive Now 进入还需使用命令定义工作文件名或使用默认的文件名，使用该方式进入一般是为恢复上一次中断的分析。所以在开始分析一个问题时，建议使用 Interactive 进入交互模式。

进入系统后会有6个窗口，提供使用者与软件之间的交流，凭借这6个窗口可以非常容易的输入命令、检查模型的建立、观察分析结果及图形输出与打印。整个窗口系统称为GUI(Graphical User Interface).如图2-1所示。

各窗口的功能如下：

1. 应用命令菜单 (Utility Menu): 包含各种应用命令，如文件控制 (File)、对象选择 (Select)、资料列式 (List)、图形显示 (Pplot)、图形控制 (PlotCtrls)、工作界面设定 (WorkPlane)、参数化设计 (Parameters)、宏命令 (Macro)、窗口控制 (MenuCtrls) 及辅助说明 (Help) 等。
2. 主菜单 (Main Menu): 包含分析过程的主要命令，如建立模块、外力负载、边界条件、分析类型的选择、求解过程等。
3. 工具栏 (Toolbar): 执行命令的快捷方式，可依照各人爱好自行设定。
4. 输入窗口 (Input Window): 该窗口是输入命令的地方，同时可监视命令的历程。
5. 图形窗口 (Graphic Window): 显示使用者所建立的模块及查看结果分析。
6. 输出窗口 (Output Window): 该窗口叙述了输入命令执行的结果。




图 2-1 ANSYS 6个窗口示意图

2.2 有限元法的基本构架

目前在工程领域内常用的数值模拟方法有：有限元法、边界元法、离散单元法和有限差分法，就其广泛性而言，主要还是有限单元法。它的基本思想是将问题的求解域划分为一系列的单元，单元之间仅靠节点相连。单元内部的待求量可由单元节点量通过选定的函数关系插值得到。由于单元形状简单，易于平衡关系和能量关系建立节点量的方程式，然后将各单元方程集组成总体代数方程组，计入边界条件后可对方程求解。

有限元的基本构成：

1. 节点（Node）：就是考虑工程系统中的一个点的坐标位置，构成有限元系统的基本对象。具有其物理意义的自由度，该自由度为结构系统受到外力后，系统的反应。
2. 元素（Element）：元素是节点与节点相连而成，元素的组合由各节点相互连接。不同特性的工程系统，可选用不同种类的元素，ANSYS提供了一百多种元素，故使用是必须慎重选则元素型号。
3. 自由度（Degree Of Freedom）：上面提到节点具有某种程度的自由度，以表示工程系统受到外力后的反应结果。要知道节点的自由度数，请查看ANSYS自带的帮助文档（Help/Element Refrence），那里有每种元素类型的详尽介绍。

2.3 ANSYS 架构及命令

ANSYS 构架分为两层，一是起始层（Begin Level），二是处理层（Processor Level）。这两个层的关系主要是使用命令输入时，要通过起始层进入不同的处理器。处理器可视为解决问题步骤中的组合命令，它解决问题的基本流程叙述如下：

1. 前置处理（General Preprocessor, PREP7）
 - 1) 建立有限元模型所需输入的资料，如节点、坐标资料、元素内节点排列次序
 - 2) 材料属性
 - 3) 元素切割的产生
2. 求解处理（Solution Processor, SOLU）
 - 1) 负载条件
 - 2) 边界条件及求解
3. 后置处理（General Postprocessor, POST1 或 Time Domain Postprocessor, POST26）

POST1 用于静态结构分析、屈曲分析及模态分析，将解题部分所得的解答如：变位、应力、反力

等资料，通过图形接口以各种不同表示方式把等位移图、等应力图等显示出来。POST26 仅用于动态结构分析，用于与时间相关的时域处理。

【例 2-1】

考虑悬臂梁如图 2-2，求 $x=L$ 变形量。已知条件：杨氏系数 $E=200E9$ ；截面参数： $t=0.01m$, $w=0.03m$, $A=3E-4$, $I=2.5E-9$ ；几何参数： $L=4m$, $a=2m$, $b=2m$ ；边界外力 $F=2N$, $q=0.05N/m$ 。




图 2-2 悬臂梁

使用 ANSYS 解决该问题的命令如下：

```

/FILNAM,EX2-1 ! 定义文件名
/TITLE,CANTILEVER BEAM DEFLECTION ! 定义分析的标题
/UNITS,SI ! 定义单位制（注意观察输出窗口的单位）
/PREP7 ! 进入前置处理
ET,1,3 ! 定义元素类型为 beam3
MP,EX,1,200E9 ! 定义杨氏模量
R,1,3E-4,2.5E-9,0.01 ! 定义实常数（要严格根据该元素类型的说明文档所给出的实常数格式）
N,1,0,0 ! 定义第 1 号节点 X 坐标为 0, Y 坐标为 0
N,2,1,0 ! 定义第 2 号节点 X 坐标为 1, Y 坐标为 0
N,3,2,0 ! 定义第 3 号节点 X 坐标为 2, Y 坐标为 0
N,4,3,0 ! 定义第 4 号节点 X 坐标为 3, Y 坐标为 0
N,5,4,0 ! 定义第 5 号节点 X 坐标为 4, Y 坐标为 0
E,1,2 ! 把 1、2 号节点相连构成单元，系统将自定义为 1 号单元
E,2,3 ! 把 2、3 号节点相连构成单元，系统将自定义为 2 号单元
E,3,4 ! 把 3、4 号节点相连构成单元，系统将自定义为 3 号单元
E,4,5 ! 把 4、5 号节点相连构成单元，系统将自定义为 4 号单元
FINISH ! 退出该处理层
/SOLU ! 进入求解处理器
D,1,ALL,0 ! 对 1 节点施加约束使它 X, Y 向位移都为 0
F,3,FY,-2 ! 在 3 节点加集中外力向下 2N
SFBEAM,3,1,PRES,0.05 ! 在 3 号元素的第 1 个面上施加压力（beam3 有四个面可通过命令 help,beam3 查看，任何一个命令都可以通过 help,命令查看帮助文档）
SFBEAM,4,1,PRES,0.05 ! 同上的在 4 号元素的第 1 个面加压力
SOLVE ! 计算求解
FINISH ! 完成该处理层
/POST1 ! 进入后处理
SET,1,1 ! 查看子步 1，在有限元中复杂的载荷可以看做简单的载荷相互叠加，在 ANSYS 中每施加一类载荷都可以进行一次求解，可以查看它对结构的影响，称为子步。
PLDISP ! 显示变形后的形状
FINISH ! 完成

```

在静态结构分析中，由 Begin Level 进入处理器，可通过斜杠加处理器的名称，如 **/prep7**、**/solu**、**/post1**。处理器间的转换通过 **finish** 命令先回到 Begin Level，然后进入想到达的处理器位置，如（图 2-3）所示。




图 2-3 处理器间的转化

2.4 典型的分析过程

ANSYS 分析过程包含三个主要的步骤：

1. 创建有限元模型
 - 1) 创建或读入有限元模型
 - 2) 定义材料属性
 - 3) 划分网格
2. 施加载荷并求解
 - 1) 施加载荷及设定约束条件
 - 2) 求解
3. 查看结果
 - 1) 查看分析结果
 - 2) 检查结果是否正确

2.5 ANSYS 文件及工作文件名

ANSYS 在分析过程中需要读写文件，文件格式为 **jobname.ext**，其中 **jobname** 是设定的工作文件名，**ext** 是由 ANSYS 定义的扩展名，用于区分文件的用途和类型，默认的工作文件名是 **file**。ANSYS 分析中有一些特殊的文件，其中主要的几个是数据库文件 **jobname.db**、记录文件 **jobname.log**、输出文件 **jobname.out**、错误文件 **jobname.err**、结果文件 **jobname.rxx** 及图形文件 **jobname.grph**。

【例 2-2】固定端杆件受到外力 **F1** 及 **F2** 的力，如图 2-4，求固定端的作用力。图（a）为实际的工程系统，图（b）为转化后的有限元模型系统，其中包含 4 个节点、3 个元素。

外力负载及约束条件为：

- 1) 第二点受外力负载 **F2**
- 2) 第三点受外力负载 **F3**
- 3) 第一点和第四点不产生任何变形（约束条件）




图 2-4 杆件

下面给出解题的 ANSYS 命令，请小心输入，注意所产生的文件。

```

/FILNAM,EX2-2  ! 定义文件名
/PREP7
ET, 1, LINK1    ! 定义杆单元
R, 1, 1         ! 定义实常数
MP, EX, 1, 30E6
N, 1
N, 2, 0, 4
N, 3, 0, 7
N, 4, 0, 10
E, 1,2    $E, 2, 3  $E, 3,4    ! 可以有"$"在一行输入多个命令
D, 1, ALL, , 4, 3    ! 在 1、4 节点施加约束
F, 2, FY, -500
F, 3, FY, -1000
SAVE          ! 存数据文件
FINISH
/SOLU
SOLVE
FINISH
EXIT

```

2.6 图形控制

图形在校验前处理的数据和后处理中检查结果者是非常重要的。ANSYS 的图形常用功能如下：

- 在实体模型和有限元模型上边界条件显示
- 计算结果的彩色等值线显示
- 可以对视图进行放大、缩小、平移、旋转等操作
- 用于实体显示的橡皮筋技术
- 多窗口显示
- 隐藏线、剖面及透视显示
- 边缘显示
- 变形比率控制

- 三维内直观化显示
- 动画显示
- 窗口背影的选择

以上功能利用 GUI 可方便实现，如打开图形控制窗口（Utility Menu>PlotCtrls>Pan>Pan,Zoom,Rotate……）可对图形进行放大、缩小、平移、旋转等操作。也可通过键盘各三键鼠标实现上操作，同时按下 Ctrl 键和鼠标左键并拖移可实现视图的平移；同时按下 Ctrl 键和鼠标中键并拖移可实现视图的缩放各 Z 向旋转（上下拖动实现缩放，左右实现旋转）；同时按下 Ctrl 键和鼠标中键并拖移可实现视图的 X 及 Y 向旋转。

3 有限元模型的建立

3.1 建模方法

由节点和元素构成的有限元模型与机械结构系统的几何外型基本是一致的。有限元模型的建立可分为直接法和间接法（也称实体模型 Solid Modeling），直接法为直接根据机械结构的几何外型建立节点和元素，因此直接法只适应于简单的机械结构系统。反之，间接法适应于节点及元素数目较多的复杂几何外型机械结构系统。该方法通过点、线、面、体积，先建立有限元模型，再进行实体网格划分，以完成有限元模型的建立。请看下面对一个平板建模的例子，把该板分为四个元素。若用直接建模法，如图 3-1，首先建立节点 1~9（如 N, 1, 0, 0），定义元素类型后，连接相邻节点生成四个元素（如 E, 1, 2, 5, 4）。如果用间接法，如图 3-2，先建立一块面积，再用二维空间四边形元素将面积分为 9 个节点及 4 元素的有限元模型，即需在网格划分时，设定网格尺寸或密度。注意用间接法，节点及元素的序号不容易控制，其节点等对象的序号的安排可能会与给定的图例存在差异。本章主要讨论直接法构建有限元模型，下一章介绍间接法（实体模型）有限元的建立。




图 3-1 直接法




图 3-2 间接法

3.2 坐标系统及工作平面

空间任何一点通常可用卡式坐标（Cartesian）、圆柱坐标（Cylinder）或球面坐标（Sphericity）来表示该点的坐标位置，不管哪种坐标系者需要三个参数来表示该点的正确位置。每一坐标系都有确定的代号，进入 ANSYS 的默认坐标系是卡式坐标系。上述的三个坐标系又称为整体坐标系，在某些情况下可通过辅助节点来定义局部坐标系。

工作平面是一个参考平面，类似于绘图板，可依用户要示移动。欲显示工作平面可用如下操作：

GUI:Utility Menu>Work Plane

GUI:Utility Menu>work Plane>Display Working Plane

欲设置平面辅助网格开关可用如下操作：

GUI:Utility Menu>Work Plane>WP Settings

相关命令

LOCAL,KCN,KCS,XC,YC,ZC,THXY,THYZ,THZX,PAR1,PAR2

定义局部坐标系，以辅助有限元模型的建立，只要在建立节点前确定用何坐标系系统即可。

KCN:坐标系统代号, 大于 10 的任何一个号码都可以。

KCS:局部坐标系统的属性。

KCS=0 卡式坐标; **KCS=1** 圆柱坐标; **KCS=2** 球面坐标;

XC,YC,ZC:局域坐标与整体坐标系统原点的关系。

THXY,THYZ,THZX:局域坐标与整体坐标系统 X、Y、Z 轴的关系。

Menu Paths: Utility Menu>WorkPlane>Local Coordinate Systems>Creat Local CS>At Specified Loc

CSYS, KSN

声明坐标系统, 默认为卡式坐标系统 (CSYS,0), **KSN** 为坐标系统代号, 1 为柱面坐标系统, 2 为球面坐标系统。

Menu Paths:Utility Menu>WorkPlane>Change Active CS to>(CSYS Type)

Menu Paths:Utility Menu>WorkPlane>Change Active CS to>Working Plane

Menu Paths:Utility Menu>WorkPlane>Offset WP to>Global Origin

/UNITS, LABEL

声明单位系统, 表示分析时所用的单位, **LABEL** 表示系统单位, 如下所示

LABEL=SI (公制, 公尺、公斤、秒)

LABEL=CSG (公制, 公分、公克、秒)

LABEL=BFT (英制, 长度=ft)

LABEL=BIN (英制, 长度=in)

3.3 节点定义

有限元模型的建立是将机械结构转换为多节点和元素相连接, 所以节点即为机械结构中一个点的坐标, 指定一个号码和坐标位置。在 ANSYS 中所建立的对象 (坐标系、节点、点、线、面、体积等) 都有编号。

相关命令

N, NODE, X, Y, Z, THXY, THYZ, THZX

定义节点, 若在圆柱坐标系统下 x, y, z 对应 r, θ, z , 在球面系统下对应 r, θ, ϕ 。

NODE: 欲建立节点的号码

X, Y, Z: 节点在目前坐标系统下的坐标位置

Menu Paths:Main Menu>Preprocessor>Create>Node>In Active CS

Menu Paths Main Menu>Preprocessor>Create>Node>On Working Plane

NDELE, NODE1, NODE2, NINC

删除在序号在 **NODE1** 号 **NODE2** 间隔为 **NINC** 的所有节点, 但若节点已连成元素, 要删除节点必先删除元素。例如:

NDELE, 1, 100, 1 ! 删除从 1 到 100 的所有点

NDELE, 1, 100, 99 ! 删除 1 和 100 两个点

Menu Paths:Main Menu>Preprocessor>Delete>Nodes

NPLOT, KNUM

节点显示, 该命令是将现有卡式坐标系统下节点显示在图形窗口中, 以供使用者参考及查看模块的建立。建构模块的显示为软件的重要功能之一, 以检查建立的对象是否正确。有限元型的建立程中, 经常会检查 各个对象的正确性及相关位置, 包含对象视角、对象号码等, 所以图形显示为有限元模型建立过程中不可缺少的步骤。**KNUM=0** 不显示号码, 为 1 显示同时显示节点号

Menu Paths:Utility Menu>plot>nodes

Menu Paths:Utility Menu>plot>Numbering... (选中 NODE 选项)

NLIST,NODE1,NODE2,NINC,Lcoord,SORT1,SORT2,SORT3

节点列式, 该命令将现有卡式坐标系统下节点的资料列示于窗口中 (会打开一个新的窗口), 使用者可检查建立的坐标点是否正确, 并可资料保存为一个文件。如欲在其它坐标系统下显示节点资料, 可以先行改变显示系统, 例如圆柱坐标系统, 执行命令 **DSYS,1**。

Menu Paths:Utility Menu>List>Nodes

FILL,NODE1,NODE2,NFILL,NSTRT,NINC,ITIME,INC,SPACE

节点的填充命令是自动将两节点在现有的坐标系统下填充许多点, 两节点间填充的节点个数及分布状态视其参数而定, 系统的设定为均分填满。**NODE1,NODE2** 为欲填充点的起始节点号码及终结节点号码, 例如两节点号码为 1 (**NODE1**) 和 5 (**NODE2**), 则平均填充三个节点 (2, 3, 4) 介于节点 1 和 5 之间。

Menu Paths:Main Menu>Preprocessor>Create>Node>Fill between Nds

NGEN,ITIME,INC,NODE1,NODE2,NINC,DX,DY,DZ,SPACE

节点复制命令是将一组节点在现有坐标系统下复制到其他位置。

ITIME: 复制的次数, 包含自己本身。

INC: 每次复制节点时节点号码的增加量。

NODE1,NODE2,NINC: 选取要复制的节点, 即要对哪些节点进行复制。

DX,DY,DZ: 每次复制时在现有坐标系统下, 几何位置的改变量。

Menu Paths:Main Menu>Preprocessor>(-Modeling-)Copy>(-Nodes-)Copy

3.4 元素的定义

当节点建立完成后, 必须使用适当元素, 将机械结构按照节点连接成元素, 并完成其有限元模型。元素选择正确与否, 将决定其最后的分析结果。ANSYS 提供了 120 多种不同性质与类别的元素, 每一个元素都有其固定的编号, 例如 **LINK1** 是第 1 号元素、**SOLID45** 是第 45 号元素。每个元素前的名称可判断该元素适用范围及其形状, 基本上元素类别可分为 1-D 线元、2-D 平面元素及 3-D 立体元素。1-D 线元素同两点连接而成, 2-D 元素由三点连成三角形或四点连成四边形, 3-D 元素可由八点连接成六面体、四点连接成角锥体或六点连接成三角柱体。每个元素的用法在 ANSYS 的帮助文档中都有详细的说明, 可用 **HELP** 命令查看。

建立元素前必须先行定义使用者欲选择的元素型号、元素材料特性、元素几何特性等, 为了程序的协调性一般在 **/PREP7** 后, 就定义元素型号及相关资料, 只要在建立元素前说明使用哪种元素即可。

相关命令

ET,ITYPE,Ename,KOPT1,KOPT2,KOPT3,KOPT4,KOPT5,KOPT6,INOPR

元素类型 (Element Type) 为机械结构系统的含的元素类型种类, 例如桌子可由桌面平面单元各桌脚梁单元构成, 故有两个元素类型。**ET** 命令是由 ANSYS 元素库中选择某个元素并定义该结构分析所使用的元素类型号码。

ITYPE: 元素类型的号码

Ename: ANSYS 元素库的名称, 即使用者所选择的元素。

KOPT1~KOPT6: 元素特性编码。

Menu Paths:Main Menu>Preprocessor Element Type>Add/Edit/Delete

MP,Lab,MAT,C0,C1,C2,C3,C4

定义材料的属性 (Material Property), 材料属性为固定值时, 其值为 **C0**, 当随温度变化时, 由后四个

参数控制。

MAT:对应 ET 所定义的号码 (ITYPE), 表示该组属性属于 ITYPE。

Lab:材料属性类别, 任何元素具备何种属性在元素属性表中均有说明。例如杨氏系数 ($Lab=EX,EY,EZ$), 密度 ($Lab=DENS$), 泊松比 ($Lab=NUXY,NUXYZ,NUZX$), 剪切模数 ($Lab=GXY,GYZ,GXZ$), 热膨胀系数 ($Lab=ALPX,ALPY,ALPZ$) 等。

Menu paths:Main Menu>Preprocessor>Matial Props>Isotropic

R,NSET,R1,R2,R3,R4,R5,R6

定义“实常数”, 即某一单元的补充几何特征, 如梁单元的面积, 壳单元的厚度。所带的参数必须与元素表的顺序一致。

Menu paths:Main Menu>Preprocessor>Real Constants

E,I,J,K,L,M,N,O,P

定义元素的连接方式, 元素表已对该元素连接顺序作出了说明, 通常 2-D 平面元素节点顺序采用顺时针逆时针均可以, 但结构中的所有元素并不一定全采用顺时针或逆时针顺序。3-D 八点六面体元素, 节点顺序采用相对应的顺时针或逆时针皆可。当元素建立后, 该元素的属性便由前面所定义的 ET,MP,R 来决定, 所以元素定义前一定要定义 ET,MP,R。I~P 为定义元素节点的顺序号码。

Menu paths:Main Menu>Preprocessor>Create>Elements>Thru Nodes

EGEN,IIME,NINC,IEL1,IEL2,IEINC,MINC,IINC,RINC,CINC

元素复制命令是将一组元素在现有坐标下复制到其他位置, 但条件是必须先建立节点, 节点之间的号码要有所关联。

ITIME:复制次数, 包括自己本身。

NINC: 每次复制元素时, 相对应节点号码的增加量。

IEL1,IEL2,IEINC: 远取复制的元素, 即哪些元素要复制。

EPLOT

元素显示, 该命令是将现有元素在卡式坐标系统下显示在图形窗口中, 以供使用者参考及查看模块。

Menu paths:Utility Menu>plot>Elements

Menu paths:Utility Menu>PlotCtrls>Numbering...

ELIST

元素列示命令是将现有的元素资料, 以卡式坐标系统列于窗口中, 使用者可检查其所建元素属性是否正确。

Menu paths:Utility Menu>List>Element>(Attributes Type)

TYPE, ITYPE

声明使用哪一组定义了的元素类型, 与 ET 命令相对应。

Menu paths:Main Menu>Preprocessor>Create>Elements>Elem Attributes

Menu paths:Main Menu>Preprocessor>Define>Default Attribs

REAL, NSET

声明使用哪一组定义了的实常数, 与 R 命令相对应。

Menu paths:同上。

MAT, MAT

使用哪一组定义了的元素属性，与 MP 命令相对应。

Menu paths:同上。

3.5 负载定义

ANSYS 中有不同的方法施加负载以达到分析的需要。负载可分为边界条件 (boundary condition) 和实际外力 (external force) 两大类，在不同领域中负载的类型有：

结构力学：位移、集中力、压力（分布力）、温度（热应力）、重力

热学：温度、热流率、热源、对流、无限表面

磁学：磁声、磁通量、磁源密度、无限表面

电学：电位、电流、电荷、电荷密度

流体力学：速度、压力

以特性而言，负载可分为六大类：DOF 约束、力（集中载荷）、表面载荷、体积载荷、惯性力有耦合场载荷。

1. DOF constraint (DOF 约束) 将给定某一自由度用一已知值。例如，结构分析中约束被指定为位移和对称边界条件；在热力学分析中指定为温度和热通量平行的边界条件。
2. Force(力) 为施加于模型节点的集中荷。如在模型中被指定的力和力矩。
3. Surface load(表面载荷) 为施加于某个面的分布载荷。例如在结构分析中为压力。
4. Body load(体积载荷) 为体积的或场载荷。在结构分析中为温度和 fluences。
5. Inertia loads(惯性载荷) 由物体惯性引起的载荷，如重力和加速度，角速度和角中速度。
6. Coupled-field loads(耦合场载荷) 为以上载荷的一种特殊情况，从一种分析得到的结果用作为另一种分析的载荷。

相关命令

/SOLU

进入解题处理器，当有限元模型建立完以后，便可以进入 /SOLU 处理器，声明各种负载。但大部分负载的载声明也可在 /PREP7 中完成，建议全部负载在 /SOLU 处理中进行声明。

/ANTYPE, Antype, Status

声明分析类型，即欲进行哪种分析，系统默认为静力学分析。

Antype=STATIC or 0 静态分析（系统默认）

BUCKLE or 1 屈曲分析

MODAL or 2 振动模态分析

HARMIC or 3 调和外力动和系统

TRANS or 4 瞬时动力系统分析

Menu Paths: Main Menu>Prprocessor>Loads>New Analysis

Menu Paths: Main Menu>Prprocessor>Loads>Restart

Menu Paths: Main Menu>Prprocessor>Solution>New Analysis

Menu Paths: Main Menu>Prprocessor>Solution>Restart

F, NODE, Lab, VALUE, VALUE2, NEND, NINC

定义节点的集中力 (Force)。

NODE: 节点号码。

Lab: 外力的形式。

Lab=FX, FY, FZ, MX, MY, MZ (结构力学的方向、力矩方向)

=HEAT (热学的热流量)

=AMP, CHRG (电学的电流、载荷)

=FLUX (磁学的磁通量)

VALUE:外力的大小。

NODE,NEND,NINC:选取施力节点的范围，故在建立节点时应先规划节点的号码，以方便整个程序的编辑。

Menu Paths:Main Menu>Solution>Apply>(Load Type)>On Node

D,NODE,Lab,VALUE,VALUE2,NEND,NINC,Lab2, ...,Lab6

定义节点自由度 (Degree of Freedom) 的限制。

NODE,NEND,NINC:选取自由度约束节点的范围。

Lab:相对元素的每一个节点受自由度约束的形式。

结构力学: DX,DY,DZ (直线位移); ROTX,ROTY,ROTZ (旋转位移)。

热 学: TEMP (温度)。

流体力学: PRES (压力); VX,VY,VZ (速度)。

磁 学: MAG (磁位能); AX,AY,AZ (向量磁位能)。

电 学: VOLT (电压)。

Menu Paths:Main Menu>Solution>Apply>(displacement type)>On Nodes

SFBEAM,ELEM,LKEY,Lab,VALI, VALJ, VAL2I, VAL2J, IOFFST, JOFFST

定义在梁元素上的分布力。

ELEM:元素号码。

LKEY:建立元素后，依节点顺序梁元素有四个面，该参为分力所施加的面号。

Lab:PRES(表示分布压力)。

VALI,VALJ:在 I 点及 J 点分布力的值。

Menu Paths:Main Menu>Solution>Apply>Plessure>On Beams

SFE,ELEM,LKEY,Lab,KVAL,VALI,VAL2,VAL3,VAL4

定义分布力作用于元素上的方式和大小，元素可分为 2-D 元素及 3-D 元素，如图 3-3 所示。VAL1~VAL4 为初建元素时节点顺序。




图 3-3 2D及3D元素分布力参数

ELEM:元素号码。

LKEY:建立元素后,依节点顺序,该分布力定义施加边或面的号码

Lab:力的形式。

Lab=PRES 结构压力

=CONV 热学的对流

=HFLUX 热学的热流率

VAL1~VAL4:相对应作用于元素边及面上节点的值。

Menu Paths:Main Menu>Solution>Apply>(load type)>(type option)

SF,Nlist,Lab,VALUE1,VALUE2

定义节点间分布力。该命令和 **SFE** 命令相似,均为定义分布力。但 **SFE** 指定特定元素分布力,作用于元素的边、面上的状态,故适用于非均匀分布力。**SF** 适用于均匀载荷,分布力作用于 **Nlist** 节点所包含元素的边及面。如图 3-4 所示。




图 3-4 SFE与SF分布力比较

Nlist:分布力作用的边或面上的所有节点。通常有 **NSEL** 命令选择节点为 **Active** 节点,然后设定 **Nlist=ALL**,表示 **Nlist** 含有 **NSEL** 所选择的所有节点。

Lab:力的形式。

Lab=PRES 结构压力

=CONV 热学的对流

=HFLUX 热学的热流率

VALUE1:作用分布力的值。

VALUE2:若 *Lab*=CONV,该值为对流的外界温度,其他领域的分析不使用该参数。

Menu Paths:Main Menu>Solution>Apply>(load type)>On Nodes

NSEL,Type,Item,Comp,VMIN,VMAX,VINC,KABS

完成有限元模型节点、元素建立后,选择对象非常重要,正常情况下在 **ANSYS** 中所建立的任何对象(节点、元素),皆为有效(**Active**)对象,只有是 **Active** 对象才能对其进行操作,为配合建模简化命令,可适时选取某些对象为 **Active** 对象,再对其进行操作。

Type:选择方式。

Type=S 选择一组节点为 **Active** 节点

=R 在现有的 **Active** 节点中,重新选取 **Active** 节点

=A 再选择某些节点,加入 **Active** 节点中

=U 在现有 **Active** 节点中,排除某些节点

=ALL 选择所有节点为 **Active** 节点

Item:资料卷标

Item=NODE 用节点号码选取

=LOC 用节点坐标选取

Comp=(无) (*Item*=NODE)

=X(YZ)(表示节点 X(YZ)为准, 当 *Item*=LOC)

VIMIN,VMAX,VINC:选取范围, *Item*=NODE 其范围为节点号码, *Item*=LOC 范围为 *Comp* 坐标的范围。
如图 (3-5) 所示。




图 3-5 选择有效的节点

3.6 求解

求解前先保存数据库, 将 Output 窗口提到最前面观察求解信息, 然后在 OLU 处理器里, 输入 SOLVE 命令即可求解。GUI 路径为 Main Menu:Solution>Solve-Current LS。如果求解失败, 典型的原因有:

- 1) 约束不够 (通常出现的问题)。
- 2) 材料性质参数有负值, 如密度值等。
- 3) 示约束铰接结构, 如两个水平运动的梁单元在竖直方向没有约束。
- 4) 屈曲—当应力刚化效应为负 (压) 时, 在载荷作用下整个结构刚度弱化。如果刚度减小到 0 或更小时, 求解存在奇异性, 因为整个结构已发生屈曲。

5) 模型中有非线性单元。

3.7 用 POST1 进行结果后处理

1. 进入 POST1

命令: **/POST1**

GUI:Main Menu>General Postproc

2. 读取结果

依据载荷步和子步号或者时间读取出需要的载荷步和子步结果。

命令: **SET**

GUI:Main Menu>General Postproc>Read Results-Load step

3. 绘变形图

命令: **PLDISP,KUND**

KUND=0 显示变形后的的结构形状

KUND=1 同时显示变形前及变形后的的结构形状

KUND=1 同时显示变形前及变形后的的结构形状，但仅显示结构外观

GUI:Main Menu>General Postprocessor>Plot Results>Deformed Shape

4. 变形动画

以动画的方式模拟结构静力作用下的变形过程

GUI:Utility Menu>Plotctrls>Animate>Deformed Shape

5. 列表支反力

在任一方向，支反力总和必等于在此方向的载荷总和

GUI:Main Menu>General Postprocessor>List Results>Reaction Solution...

6. 应力等值线与应力等值线动画

应力等值线方法可清晰描述一种结果在整个模型中的变化，可以快速确定模型中的危险区域。

GUI:Main Menu>General Postprocessor>Plot Results>-Contour Plot-Nodal Solution...

应力等值线动画

GUI:Utility Menu>Plotctrls>Animate>Deformed Shape

【例 3-1】建一个平面结构节点的安排图，如图 3-6 所示




图 3-6 例3-1图

/FILENAME,EX3-1

/UNITS,BIN

/TITLE,PLANE NODES GENERATION

/PREP7

```

LOCAL,11,1,5,0,0      ! 建立 11 号局部圆柱区域坐标
N,1,5,30
N,2,5,60
CSYS,0                 ! 回至卡式坐标
N,3,0,5
N,4,2.5,-5
CSYS,11               ! 回至 11 号圆柱坐标
N,5,5,0
N,6,0,0
N,7,5,-45
CSYS,0
N,8,5,5
N,9,7.5,0
.....

```

【例 3-2】

有一个梁结构如图 3-7 (a) 所示, $L=30\text{cm}$, $F=1000\text{N}$, $q=600\text{N/m}$ 图 3-6 (b) 为均布 11 个节点的规划。




图 3-7 例3-2图

解题的 ANSYS 命令如下

```

/FILNAME,EX3-2
/UNITS,SI
/TITLE,BEAM 11 NODE SIMULATION
/PREP7
N,1,0,0
N,11,0.3,0
FILL,1,11
ET,1,BEAM3
MP,EX,1,207E9
R,1,1e-4,2.083e-10,0.005
E,1,2
EGEN,10,1,1,1,1
EPLLOT

```



```

/PNUM,ELEM,1
EPLOT
FINISH
/SOLU
ANTYPE,STATIC
OUTPR,BASIC,ALL      ! 在输出窗口中列出元素的结果
D,1,UX,0,,,,,UY,ROTZ
D,11,UX,0,,,,,UY
SFBEAM,1,1,PRES,0,60
SFBEAM,2,1,PRES,60,120    $ SFBEAM,3,1,PRES,120,180
SFBEAM,4,1,PRES,180,240    $ SFBEAM,5,1,PRES,240,300
SFBEAM,6,1,PRES,300,360    $ SFBEAM,7,1,PRES,360,420
SFBEAM,8,1,PRES,420,480    $ SFBEAM,9,1,PRES,480,540
SFBEAM,10,1,PRES,540,600
F,6,FY,-1000
SOLVE
FINISH
/POST1
PLDISP                ! 显示变形图
PRDISP                ! 列出变形资料
FINISH

```

4 实体模型的建立

4.1 实体模型简介

在上一章里已介绍了有限的直法建模，但该方法对复杂的结构，建立过程复杂而且容易出错，因此这里引入实体模型的建立，与一般的 CAD 软件一样，利用点、线、面、体积组合而成。实体模型几何图形决定之后，由这界来决定网格，即每一线段要分成几个元素或元素的尺寸是多大。决定了每边元素数目或尺寸大小之后，ANSYS 的内建程序即能自动产生网格，即自动产生节点和元素，并同时完成有限元模型。

4.2 实体模型的建立方法

实体模型建立有下列方法：

1) 由下往上法 (bottom-up Method)

由建立最低单元的点到最高单元的体积，即建立点，再由点连成线，然后由线组合成面积，最后由面积组合建立体积。

2) 由上往下法 (top-down method) 及布尔运算命令一起使用

此方法直接建立较高单元对象，其所对应的较低单元对象一起产生，对象单元高低顺序依次为体积、面积、线段及点。所谓布尔运算为对象相互加、减、组合等。

3) 混合使用前两种方法

依照个人的经验，可结合前两种方法综合运用，但应考虑到要获得什么样的有限元模型，即在网格化分时，要产生自由网格划分或对应网格划分。自由网格划分时，实体模型的建立比较简单，只要所有的面积或体积能接合成一个体就可以，对应网格划分时，平面结构一定要四边形或三边面积相接而成，立体结构一定要六面体相接而成。

4.3 群组命令介绍

表 4-1 给出了 ANSYS 中 X 对象的名称，表 4-2 中列出了 ANSYS 中 X 对象的群组命令，命令参数大部分与节点及元素相似。以后对组命令不再详述。

表 4-1 ANSYS 中 X 对象的名称

对象种类 (X)	节点	元素	点	线	面积	体积
对像名称	X=N	X=E	X=K	X=L	X=A	X=V

表 4-2 ANSYS 中 X 对象的群组命令

群组命令	意 义	例 子
XDELE	删除 X 对象	LDELE 删除线
XLIST	在窗口中列示 X 对象	VLIST 在窗口中列出体积资料
XGEN	复制 X 对象	VGEN 复制体积
XSEL	选择 X 对象	NSEL 选择节点
XSUM	计算 X 对象几何资料	ASUM 计算面积的几何资料，如面积大小、边长、重心等
XMESH	网格化 X 对象	AMESH 面积网格化 LMESH 线的网格化
XCLEAR	清除 X 对象网格	ACLEAR 清除面积网格 VCLEAR 清除体积网格
XPLOT	在窗口中显示 X 对象	KPLOT 在窗口中显示点 APLOT 在窗口中显示面积

4.4 点定义

实体模型建立时，点是最小的单元对象，点即为机械结构中一个点的坐标，点与点连接成线也可直接组合成面积及体积。点的建立按实体模型的需要而设定，但有时会建立些辅助点以帮助其它命令的执行，如圆弧的建立。

相关命令：

K,NPT,X,Y,Z

建立点（Keypoint）坐标位置（X,YZ）及点的号码 NPT 时，号码的安排不影响实体模型的建立，点的建立也不一要连号，但为了数据管理方便，定义点之前先规划好点的号码，有利于实体模型的建立。在圆柱坐标系下，X,YZ 对应于 R， θ ，Z，球面坐标下对应着 R， θ ， Φ 。

Menu Paths:Main Menu>Preprocessor>Create>Key Point>In Active Cs

Menu Paths:Main Menu>Preprocessor>Create>Key Point>On Working Plane

KFILL,NP1,NP2,NFILL,NSTRT,NINC,SPACE

点的填充命令是自动将两点 NP1,NP2 间，在现有的坐标系下填充许多点，两点间填充点的个数（NFILL）及分布状态视其参数（NSTRT,NINC,SPACE）而定，系统设定为均分填充。如语句 **FILL,1,5**，则平均填充 3 个点在 1 和 5 之间。如图 4-1 所示。



图 4-1 点填充

Menu Paths:Main Menu>Preprocessor>Create>Key Point>Fill

KNODE,NPT,NODE

定义点（NPT）于已知节点上

Menu Paths:Main Menu>Preprocessor>Create>Keypoint>On Node

4.5 线段定义

建立实体模型时，线段为面积或体积的边界，由点与点连接而成，构成不同种类的线段，例如直线、曲线、BSPLIN、圆、圆弧等，也可直接由建立面积或体积而产生。线的建立与坐标系统有关，直角坐标

系为直线，圆柱坐标下曲线。

相关命令

L,*P1,P2,NDIV,SPACE,XV1,YV1,ZV1,XV2,YV2,ZV2*

此命令是用两个点来定义线段，此线段的形状可为直线或曲线，此线段在产生面积之前可做任何修改，但若已成为面积的一部分，则不能再做任何改变，除非先把面积删除。**NDIV** 指欲进行网格化时所分的元素数目。

Menu Paths:Main Menu>Preprocessor>Create>Lines>In Active Coord

LDIV,NL1,RATIO,PDIV,NDIV,KEEP

此命令是将线分割成数条线，**NL1** 为线段的号码，**NDIV** 为线段欲分的段数（系统默认为两段），在于 2 时为均分，**RATIO** 为两段的比例（**NDIV**=2 时才起作用），**KEEP**=0 时原线段资料将删除，**KEEP**=1 则保留。

Menu Paths:Main Menu>Preprocessor>Operate>Divide>(type options)

LFILLT, NL1, NL2, RAD, PCENT

此命令是在两条相交的线段（**NL1,NL2**）间产生一条半径等于 **RAD** 的圆角线段，同是自动产生三个点，其中两个点在 **NL1,NL2** 上，是新曲线与 **NL1,NL2** 相切的点，第三个点是新曲线的圆心点（**PCENT**，若 **PENT**=0 则不产生该点），新曲线产生后原来的两条线段会改变，新形成的线段和点的号码会自动编排上去。如图 4-2 所示。




图 4-2 产生圆角

Menu Paths:Main Menu>Preprocessor>Create>Line Fillet

LARC,P1,P2,PC,RAD

定义两点（**P1,P2**）间的圆弧线（Line of Arc），其半径为 **RAD**，若 **RAD** 的值没有输入，则圆弧的半径直接从 **P1,PC** 到 **P2** 自动计算出来。不管现在坐标为何，线的形状一定是圆的一部分。**PC** 为圆弧曲率中心部分任何一点，不一定是圆心。如图 4-3 所示。



图 4-3 圆弧的生成

Menu Paths:Main Menu>Preprocessor>Create>Arcs>By End KPs & Rad

Menu Paths:Main Menu>Preprocessor>Create>Arcs>Through 3 Kps

CIRCLE, PCENT, RAD, PAXIS, PZERO, ARC, NSEG

此命令会产生圆弧线（CIRCLE Line），该圆弧线为圆的一部分，依参数状况而定，与目前所在的坐标系统无关，点的号码和圆弧的线段号码会自动产生。

PCENT: 圆弧中心点坐标号码

PAXIS: 定义圆心轴正向上任意点的号码。

PZERO: 定义圆弧线起点轴上的任意点的号码，此点不一定在圆上。

RAD : 圆的半径，若此值不输，则半径的定义为 PCENT 到 PZERO 的距离

ARC : 弧长（以角度表示），若输入为正值，则由开始轴产生一段弧长，若没输和，产生一个整圆。

NSEG : 圆弧欲划分的段数，此处段数为线条的数目，非有限元网格化时的数目。

Menu Paths: Main Menu>Preprocessor>Create>Arcs>By End Cent & Radius

Menu Paths: Main Menu>Preprocessor>Create>Arcs>Full Circle

【例 4-1】练习点和线段的生成

```

/REP7
K,1,5,4          ! 建立点 1 坐标 (5, 4)
K,4,-1,2         ! 建立点 4 坐标 (-1, 2)
K,PLOT           ! 显示点，无号码
/PNUM,KP,1
K,PLOT           ! 显示点，无号码
KLIST            ! 列出点的资料
K,,2,-2          ! 建立点 2 坐标 (2, -2), 点 2 是自动获得的最小号码
DSYS,1           ! 改变显示坐标系统为圆柱坐标
KLIST
DSYS             ! 回复显示卡式坐标系统
K,2,-3           ! 改变 2 点的坐标为 (0, -3)
CSYS,1           ! 改变坐标系统为圆柱坐标
K,,4             ! 建立点 3 坐标，半径=4，角度=0
K,4,4,30         ! 建立点 4 坐标，半径=4，角度=30
K,,4,60          ! 建立点 5 坐标，半径=4，角度=60
KLIST
DSYS,1           ! 改变显示坐标系统为圆柱坐标
KLIST
CSYS             ! 回复坐标系统为卡式坐标
DSYS             ! 回复显示坐标系统为卡式坐标
L,3,5            ! 建立点 3 至点 5 的直线段
L,PLOT           ! 显示线段，无号码
/PNUM,LINE,1
L,POT            ! 显示线段，有号码
LLIST            ! 列出线段资料
L,2,3            ! 建立点 2 至点 3 的直线段
CSYS,1           ! 改变坐标系统为圆柱坐标系统
L,2,5            ! 建立点 2 至点 5 的圆柱坐标线段
L,PLOT           ! 显示线段

```

【例 4-2】练习 LARC 命令产生圆弧线段

```

/REP7
K,1,0

```

```

K,2,1,2
K,3,1,-1
/PNUM,KP,1
/PNUM,LINE,1
KPLOT
LARC,1,2,3,2      ! 建立点 1 至点 2 的圆弧, 半径为 2
LARC,1,2,3,4      ! 建立点 1 至点 2 的圆弧, 半径为 4
LARC,1,2,3,-2     ! 建立点 1 至点 2 的圆弧, 半径为 2, 反曲率

```

【例 4-3】练习 CIRCLE 命令产生圆弧线段

```

/PREP7
K,1
K,2,3
K,3,0,3
K,4,0,0,3
/PNUM,KP,1
/PNUM,LINE,1
KPLOT
CIRCLE,1,2
CIRCLE,1,1.5,3,4,135,4 ! 产生一个 X-Z 平面, 135°, 4 段的圆弧

```

4.6 面积定义

实体模型建立时, 面积为体积的边界, 由经连接而成, 面积的建立可由点直接相接或线段围接而成, 并构成不同数目边的面积。也可直接建构体积而产生面积, 如要进行对应网格化, 则必须将实体模型建构为四边形面积的组合, 最简单的面积为 3 点连接面成, 以点围成面积时, 点必须以顺时针或逆时针输入, 面积的法向按点的顺序依右手定则决定。

相关命令

A,P1,P2,P3,P4,P5,P6,P7,P8,P9

此命令用已知的一组点 (P1~P9) 来定义面积 (Area), 最少使用三个点才能围成面积, 同时产生转围绕些面积的线段。点要依次序输入, 输入的顺序会决定面积的法线方向。如果此面积超过了四个点, 则这些点必须在同一个平面上。如图 4-4 所示。




图 4-4 由点生成面积

Menu Paths: Main Menu>Preprocessor>Create>Arbitrary>Through KPs

AL,L1,L2,L3,L4,L5,L6,L7,L8,L9,L10

此命令由已知的一组直线 (L1, ...L10) 线段 (Lines) 围绕而成的面积 (Area), 至少须要 3 条线段才能形成平面, 线段的号码没有严格的顺序限制, 只要它们能完成封闭的面积即可。同时若使用超过 4 条线段去定义平面时, 所有的线段必须在同一平面上, 以右手定则来决定面积的方向。如图 4-5 所示。

Menu Paths: Main Menu>Preprocessor>Create>Arbitrary>By Lines




图 4-5 线拉伸成面

AROTAT,NL1,NL2,NL3,NL4,NL5,NL6,PAX1,PAX2,ARC,NSEG

建立一组圆柱型面积 (Area), 产生方式为绕着某轴 PAX1, PAX2 为轴上的任意两点, 并定义轴的方向), 旋转一组已知线段 (NL1~NL6), 以已知线段为起点, 旋转角度为 ARC, NSEG 为在旋转角度方向可分的数目。如图 4-6 所示。




图4-6 由线旋转成面

Menu Paths: Main Menu>Preprocessor>Operator>Extrude/Sweep>About Axis

4.7 体积定义

体积为对象的最高单元, 最简单体积定义为点或面积组合而成。由点组合时, 最多由八点形成六面积, 八点顺序为相应面顺时针或逆时针皆可, 其所属的面积、线段, 自动产生。以面积组合时, 最多为十块面积围成的封闭体积。也可由原始对象 (Primitive Object) 建立, 例如: 圆柱、长方体、球体等可直接建立。

相关命令:

V,P1,P2,P3,P4,P5,P6,P7,P8

此命令由已知的一组点 (P1~P8) 定义体积 (Volume)，同时也产生相对应的面积及线。点的输入必须依连续的顺序，以八点面言，连接的原则为相对应面相同方向，对于四点角锥、六点角柱的建立都适用。如图 4-7 所示。




图 4-7 由点生成体

Menu paths: Main Menu>Preprocessor>Create>Arbitrary>Through KPs

VA,A1,A2,A3,A4,A5,A6,A7,A8,A9,A10

定义由已知的一组面 (VA1~VA10) 包围成的一个体积，至少需要 4 上面才能围成一个体积，些命令适用于当体积要多于 8 个点才能产生时。平面号码可以是任何次序输入，只要该组面积能围成封闭的体积即可。

Menu Paths: Main Menu>Preprocessor>Create>Arbitrary>By Areas

Menu Paths: Main Menu>Preprocessor>Create>Volume by Areas

Menu Paths:Main Menu>Preprocessor>Geom Repair>Create Vlume

VDRAG,NA1,NA2,NA3,NA4,NA5,NA6,NLP1,NLP2,NLP3,NLP4,NLP5,NLP6

体积（Volume）的建立是由一组面积（NA1~NA6），延某组线段（NL1~NL6）为路径，拉伸而成。




图 4-8 由面拉伸成体

Menu Paths:Main Menu>Operate>Extrude/Sweep>Along Lines

VROTAT,NA1,NA2,NA3,NA4,NA5,NA6,PAX1,PAX2,ARC,NSEG

建立柱形体积，即将一组面（NA1~NA6）绕轴 PAX1,PAX2 旋转而成，以已知面为起点，ARC 为旋转的角度，NSEG 为整个旋转角度中欲分的数目。




图 4-9 由面旋转生成体积

Menu Paths:Main Menu>Operate>Extrude/Sweep>About Axis

【例 4-4】综合点、线、面、体积练习

```

/PREP7
/PNUM,KP,1          $/PUM,LINE,1
/PNUM,AREA,1        $/PNUM,VOLU,1
K,1,2              ! 建立点 1，坐标（2， 0）
K,2,3,4            ! 建立点 2，体坛（3， 4）
K,3,-0.5,3         ! 建立点 3，体坛（-0.5， 3）
K,4,-2,0.5         ! 建立点 4，体坛（-2， 0.5）
A,1,2,3,4          ! 由点 1， 2， 3， 4 生成面
CSYS,1             ! 转为圆柱坐标系统
K,5,4              ! 建立点 5，半径=4，角度=0
    
```



```

K,6,4,45      ! 建立点 6，半径=4，角度=45
K,7,4,45,5    ! 建立点 7，半径=4，角度=45，高度=5
K,8,4,,5      ! 建立点 7，半径=4，角度=0，高度=5
A,5,6,7,8     ! 生成一个中心角为 45 度的圆柱面
ADEL,1        ! 删除面 1
LDEL,3        ! 删除线 1
A,1,2,3,4     ! 生成面
ADEL,2
A,5,6,8
A,8,5,6
CSYS          ! 转到卡式坐标
K,,7,1,3
/PNUM,KP,1
V,8,5,6,9
    
```

4.8 用体素创建 ANSYS 对象

这里先引入体素（Primitive）的概念，ANSYS 中，体素指预先定义好的具有共同形状的面或体。利用它可直接建立某些形状的高级对象，例如矩形、正多边形、圆柱体、球体等，高级对象的建立可节省很多时间，其所对应的低级对象同时产生，系统并给予最小的编号。我们用体素创建对象时，通常要结合一定的布你操作才能完成实体模型的建立。常用的 2-D 及 3-D 体素如下图 4-10 所示




图 4-10 ANSYS 中的体素

在创建对象时，要注意的是，3-D 对象具有高度的，其高度必须在 Z 轴方向，如欲在非原点坐标建立 3-D 体素对象，必须移动坐标平面至所需的点上，对象的高度非 Z 轴的，必须旋转工作平面。图 4-11 为一个空心球示意图，当命令 Menu paths:Main Menu>Preprocessor>Create>Sphere>By Dimensions 执行完后，其中的 4 个面、10 条线及 8 个关键点自动产生。




图 4-11 创建空心球

相关命令**RECTNG,X1,X2,Y1,Y2**

建立一长方形面积，以个对顶的坐标为参数即可。X1,X2 为 X 方向的最小及最大值，Y1, Y2 为 Y 方向的最小及最大值。

Menu paths: Main Menu>Preprocessor>Create>Rectangle>By Dimensions

PCIRC,RAD1,RAD2,THETA1,THETA2

以工作平面的坐标为基准，建立平面圆面积。RAD1,RAD2 为内外圆半径，THETA1,THETA2 为圆面的角度范围。系统默认为 360 度，并以 90 度自行分段。如图 4-12 所示




图 4-12 建立圆面

Menu paths: Main Menu>Preprocessor>Create>By Dimensions

BLOCK,X1,X2,Y1,Y2,Z1,Z2

建立一个长方体，以对顶角的坐标为参数。，X1,X2 为 X 向最小及最大坐标值，Y1,Y2 为 Y 向最小及最大坐标值，Z1,Z2 为 X 向最小及最大坐标值。

Menu paths: Main Menu>Preprocessor>Create>Block>By Dimensions

CYLIND,RAD1,RAD2,Z1,Z2,THETA1,THETA2

建立一个圆柱体积，圆柱的方向为 Z 方向，并由 Z1, Z2 确定范围，RAD1,RAD2 为圆柱的内外半径，THETA1,THETA2 为圆柱的始、终结角度，参阅 PCIRC 命令的图。

Menu paths: Main Menu>Preprocessor>Create>Cylinder>By Dimensions

4.9 布尔操作

布尔操作可对几何图元进行布尔计算, ANSYS 布尔运算包括 ADD(加), SUBTRACT(减), INTERSECT

(交)，DIVIDE（分解），GLUE（粘接），OVERLAP（搭接），它们不仅适用于简单的图元，也适用于从 CAD 系统传入的复杂几何模型。GUI 命令路径为 Main Menu>Preprocessor>-Modeling-Operate。通常情况下，结构进对应网格化几乎无法达到，故皆以自由网格化为主。同时布尔运算对所操作的对象进行编号。

【例 4-5】练习，建立图 4-13 的模型。

```

/PREP7
RECTNG,0,6,-1,1
PCIRC,0,1,90,270
RECTNG,4,6,-3,-1
WPAVE,5,-3
PCIRC,0,1,-180,0
ADD,ALL
PCIRC,0,4
WPAVE,0,0,0
PCIRC,0,4
ASBA,5,1
ASBA,3,2
    
```




图 4-13 例4-5图

5 网格划分

5.1 区分实体模型和有限元模型

现今所有的有限元分析都用实体建模，类似于 CAD，ANSYS 以数学的方式表达结构的几何形状，用于在里面填充节点和单元，还可以在几何边界上方便的施加载荷，但是几何实体模型并不参与有限元分析，所有施加在有限元边界上的载荷或约束，必须最终传递到有限元模型上（节点和单元）进行求解。参见图 5-1。




图 5-1 实体模型与有限元模型比较

5.2 网格化的步骤

1. 建立选取元素数据
2. 设定网格建立所需的参数
3. 产生网格

第一步是建立元素的数据，这些数据包括元素的种类（TYPE），元素的几何常数（R），元素的材料性质（MP），及元素形成时所在的坐标系统，也就是说当对象进行网格化分后，元素的属性是什么。当然我们可以设定不同种类的元素，相同的元素又可设定不同的几何常数，也可以设定不同的材料特性，以及不同的元素坐标系统。

第二步即可进行设定网格划分的参数，最主要是定义对象边界元素的大小和数目。网格设定所需的参数，将决定网格的大小、形状，这一步非常重要，将影响分析时的正确性和经济性。网格细也许会得到很好的结果，但并非网格划分的越细，得到的结果就越好，因为网太密太细，会占用大量的分析时间。有时

较细的网格与较粗的网格比较起来，较细的网格分析的精确度只增加百分之几，但占用的计算机资源比起较粗的网格确是数倍之多，同时在较细的网格中，常会造成不同网格划分时连接的困难，这一点不能不特别注意。

完成前两步即可进行网格划分，并完成有限元模型的建立，如果不满意网格化的结果，也可清除网格化，重新定义元素的大小、数目，再进行网格化，直到得到满意的有限元结果为止。

实体模型的网格化可分为自由网格化（Free Meshing）及对应网格化（Mapped Meshing）两种不同的网格化，对于建构实体模型过程有相当大的影响。自由网格化时实体模型的构建简单，无较多限制。反之，对应网格化，实体模型的建立比较复杂，有较多限制。

网格化 ANSYS 的基本流程如下：

```

/PREP7
ET,1
MP,EX,1
R,1
ET,2
MP,EX,2
R,2
ET,3
MP,EX,2
R,2
ET,3
MP,EX,3
R,3
! 建立实体模型
K, .....
L, .....
A, .....
V, .....
! 声明元素大小、形状及网格种类
LESIZE, .....
KESIZE, .....
ESIZE, .....
SMRTSIZE, .....
MSHKEY, .....
MSHAPE, .....
! 进行网格化
XATT,1,1,1
XMESH      ! X 对象网格化后，元素属性由 XATT 决定
XMATT,2,1,2
XMESH

```

5.3 元素形状定义

元素形状在 2-D 结构中可分为四边形和三角形，在 3-D 结构中可分为六面体和角锥体。当实体模型进行对应网格划分时，2-D 及 3-D 结构所产生的元素必为四边形及六面体，当无法进行对应网格化时，程序会自动用自由网格化，所以 2-D 结构将自行以四边形和三角形的混合方式进行，3-D 结构以角锥体方式进行。网格化，有默认尺寸大小，也就是说不给定线段和网格数目，仍然可以进行网格划分，但不一定能满

足设计者的要求。

元素大小基本上在线段上定义，可用线段数目和线段长度来划分，通常以线段数目分割比较方便。分割时可采用均分或不均分，不均分以线段方向或中间为准，根据数定义可得到渐增或渐减的效果。除此之外，也可以以整体对象为基准，确定网格的大小。此外在自由网格化一般不需要定义线段的数目及大小，程序将提供智能化控制（SMRTSIZE）；而指定线段进行元素数目及大小的声明，大多用于对应网格化。

5.4 网格划分工具

网格划分工具是网格控制的一种快捷方式，它能方便的实现单元属性控制、智能网格划分控制、尺寸控制、自由网格划分和对应网格划分、执行网格划分、清除网格划分以及局部细分。参见图 5-2。



图 5-2 网格划分工具

程序默认为自由网格划分，元素形状以四边形、六面体为准优先，三角形、角锥次之，网格化时，如果实体模型能够对应网格化，而且相对应边长度不是差的很多，则必以以应网构化优先考虑进行。

网格划分工具中，我们一般只用它的一两组功能，即可达到要求。这里有必要知道尺寸控制的优先级。缺省单元尺寸控制：

- 对线划分的指定被最先考虑
- 关键点附近的单元尺寸作为第二级考虑对象
- 总体单元尺寸作为第三级考虑对象
- 缺省尺寸最后考虑

智能单元尺寸的优先顺序

- 对线划分的指定被最先考虑
- 关键点附近的单元尺寸作为第二级考虑对象，当考虑到小的几何特征和曲率时，可以忽略它
- 总体单元尺寸作为第三级考虑对象，当考虑到小的几何特征和曲率时，可以忽略它

● 智能单元尺寸设置最后考虑

【例 5-1】综合练习，如图 5-3 所示

```

/PREP7
ET,1,PLANE42
K,1,-2.5      $K,2,6,-2.5      $K,3,,2.5      $K,4,6,2.5
CSYS,1
K,5,10,-30    $K,6,10,30
CSYS,0
SAVE
L,1,2          $L,4,3
CSYS,1,
L,2,4          $L,5,6
CSYS,0
A,1,2,4,3
A,2,5,6,4
SAVE

```

！可以有 RESUME 命令回复到当前点

AMESH,ALL ! 此时网格以系统默认的尺寸进行自由划分，但由于面积符合对应网格化的要求，所以会进行对应网划分。在网格化后，可用 RESUME 命令回到划分前的状态，试使用网格划分工具，指定不同的控制，观察结果。




图 5-3 例 5-1图

6 实体模型的外力

对于实体模型而言，施力方式可采用直接施加在实体模型的点、线、面上，施加方式同第 3 章里在节点和元素上直接施加非常相似。在实体模型上加载独立于有限元网格，重新划分网格或局部修改网格不会影响载荷，加载更加容易，可以在图形中拾取。

相关命令：

KD,KPOI,Lab,VALUE,VALUE2,KEXPND,Lab2,Lab3,Lab4,Lab5,Lab6

该命令与 D 命令相对应，定义约束，KPOI 为受限点的号码，VALUE 为受约束点的值。Lab!~Lab6 与 D 相同，可借着 KEXPND 去扩展定义在不同点间节点所受约束。

DL,LINE,AREA,Lab,Value!,value2

在线段上定义约束条件 (Displacement) LINE,AREA 为受约束线段及线段所属面积的号码。Lab 与 D 命令相同,但增加了对称 (Lab=SYMM) 与反对称 (Lab=ASYM), Value 为约束的值。

Menu paths:Main Menu>Solution>Apply>On Lines

Menu paths:Main Menu>Solution>Apply>Boundary>On Lines

Menu paths:Main Menu>Solution>Apply>Displacement>On Lines

DA, AREA,Lab,Value1,Value2

在面积上定义约束条件, AREA 为受约束的号码, Lab 同 DL。

Menu paths:Main Menu>Solution>Apply>On Areas

Menu paths:Main Menu>Solution>Apply>Boundary>On Areas

Menu paths:Main Menu>Solution>Apply>Displacement>On Areas

FK,KPOI,Lab,VALUE1,VALUE2

该命令与 F 命令相对应,在点 (Keypoint) 上定义集中外力 (Force),KPOI 为受上力点的号码, VALUE 为外力的值。Lab 与 F 命令相同

Menu paths:Main Menu>Solution>Apply>Excitation>On Keypoint

Menu paths:Main Menu>Solution>Apply>Others>On Keypoint

SFL,LINE,Lab,VALI,VALJ,VAL2I,VAL2J

该命令与 SFE 相对应,在面积线上定义分布力作用的方式和大小,应用于 2-D 的实体模型表面力。LINE 为线段的号码, Lab 的定义与 SFE 相同, VALI~VALJ 为当初建立线段时点顺序的分布力值。如图 6-1 所示。

Menu paths:Main Menu>Solution>Apply>Excitation>On Lines

Menu paths:Main Menu>Solution>Apply>Others>O On Lines




图 6-1 SFL命令示例

SFA,AREA,LKEY,Lab,VALUE1,VALUE2

该命令与 SFE 相对应,在体积的面上定义分布力作用的方式和大小,应用于 3-D 的实体模型表面力。AREA 为面积的号码, LKEY 为当初建立体积时面积的顺序,选择 AREA 与 LKEY 其中的一个输入。Lab 的定义与 SFE 相同, VALUE 为分布力的值。

Menu paths:Main Menu>Solution>Apply>Excitation>On Areas

Menu paths:Main Menu>Solution>Apply>Others>O On Areas

【例 6-1】如图 6-2 所示, $E=30e6$, 两端压力 100, 中心孔内线压分布力 500 向外, 取对称进行分析。ANSYS 解题命令如下:

```
/PREP7
ET,1,PLANE42
```




图 6-2 例 6-1图

```
MPEX,1,30E6
K,1      $K,2,15    $K,3,15,2
K,7,,10  $K,8,15,5,1
L,1,2    $L,2,3     $L,5,6
L,6,7    $L,7,1
CIRCLE,4,3,8,5,180
NUMMRG,KP          ! 合并重合点
AL,ALL
! 几何模型创建完毕
```

```
ESIZE,,4          ! 指定线段要分成的元素的数目为 4，可以用网格划分工具方便实现
AMESH,ALL
FINISH
! 网格划分完毕
/SOLU
LSEL,S,LINE,5      ! 只指定线段 5 有效
NSLL,S,1           ! 指定线段 5 上的所有节点有效
SF,ALL,PRES,100,0  ! 施中压力
ALLSEL             ! 选中所有对象有效
LSEL,S,LINE,6,7    ! 只指定线段 6 有效
NSLL,S,1           ! 线段 6 上的所有节点有效
SF,ALL,PRES,500    ! 施加压力
ALLSEL
! 在执行下一步前，可能先要设置一下间隙误差为 0.05，参见图 6-3
NSEL,S,LOC,X,14.99,15.01 ! 只指定 X 坐标在 14.99—15.01 范围内的有效
DSYM,SYMM,X        ! 对称约束
ALLSEL
SOLVE
FINISH
! 求解完毕
/POST
```


/PLDISP,1 ! 显示变形图
 /PLNSOL,S,EQV ! 显示等应力线图，如图 6-4。
 /PRNSOL ! 列出节点应力
 ! 检查结果





图 6-3 变形图




图 6-4 等应力线图

附表

结构静力学中常用的单元类型

类别	形状和特性	单元类型
杆	普通 双线性	LINK1, LINK8 LINK10
梁	普通 截面渐变 塑性 考虑剪切变形	BEAM3, BEAM4 BEAM54, BEAM44 BEAM23, BEAM24 BEAM188, BEAM189
管	普通 浸入 塑性	PIPE16, PIPE17, PIPE18 PIPE59 PIPE20, PIPE60
2-D 实体	四边形 三角形 超弹性单元 粘弹性 大应变 谐单元 P 单元	PLANE42, PLANE82, PLANE182 PLANE2 HYPER84, HYPER56, HYPER74 VISCO88 VISO106, VISO108 PLANE83, PPNAE25 PLANE145, PLANE146
3-D 实体	块 四面体 层 各向异性 超弹性单元 粘弹性 大应变 P 单元	SOLID45, SOLID95, SOLID73, SOLID185 SOLID92, SOLID72 SOLID46 SOLID64, SOLID65 HYPER86, HYPER58, HYPER158 VISO89 VISO107 SOLID147, SOLID148
壳	四边形 轴对称 层 剪切板 P 单元	SHELL93, SHELL63, SHELL41, SHELL43, SHELL181 SHELL51, SHELL61 SHELL91, SHELL99 SHELL28 SHELL150